

NAG Fortran Library Routine Document

C02AJF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

C02AJF determines the roots of a quadratic equation with real coefficients.

2 Specification

```
SUBROUTINE C02AJF(A, B, C, ZSM, ZLG, IFAIL)
INTEGER          IFAIL
real            A, B, C, ZSM(2), ZLG(2)
```

3 Description

The routine attempts to find the roots of the quadratic equation $az^2 + bz + c = 0$ (where a , b and c are real coefficients), by carefully evaluating the 'standard' closed formula

$$z = \frac{-b \pm \sqrt{b^2 - 4ac}}{2a}.$$

It is based on the routine QDRTC from Smith (1967).

Note: it is not necessary to scale the coefficients prior to calling the routine.

4 References

Smith B T (1967) ZERPOL: A zero finding algorithm for polynomials using Laguerre's method *Technical Report* Department of Computer Science, University of Toronto, Canada

5 Parameters

- | | | |
|----|--|---------------|
| 1: | A – <i>real</i> | <i>Input</i> |
| | <i>On entry:</i> A must contain a , the coefficient of z^2 . | |
| 2: | B – <i>real</i> | <i>Input</i> |
| | <i>On entry:</i> B must contain b , the coefficient of z . | |
| 3: | C – <i>real</i> | <i>Input</i> |
| | <i>On entry:</i> C must contain c , the constant coefficient. | |
| 4: | ZSM(2) – <i>real</i> array | <i>Output</i> |
| | <i>On exit:</i> the real and imaginary parts of the smallest root in magnitude are stored in ZSM(1) and ZSM(2) respectively. | |
| 5: | ZLG(2) – <i>real</i> array | <i>Output</i> |
| | <i>On exit:</i> the real and imaginary parts of the largest root in magnitude are stored in ZLG(1) and ZLG(2) respectively. | |

6: IFAIL – INTEGER

Input/Output

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, A = 0. In this case, ZSM(1) contains the root $-c/b$ and ZSM(2) contains zero.

IFAIL = 2

On entry, A = 0 and B = 0. In this case, ZSM(1) contains the largest machine representable number (see X02ALF) and ZSM(2) contains zero.

IFAIL = 3

On entry, A = 0 and the root $-c/b$ overflows. In this case, ZSM(1) contains the largest machine representable number (see X02ALF) and ZSM(2) contains zero.

IFAIL = 4

On entry, C = 0 and the root $-b/a$ overflows. In this case, both ZSM(1) and ZSM(2) contain zero.

IFAIL = 5

On entry, b is so large that b^2 is indistinguishable from $b^2 - 4ac$ and the root $-b/a$ overflows. In this case, ZSM(1) contains the root $-c/b$ and ZSM(2) contains zero.

If IFAIL > 0 on exit, then ZLG(1) contains the largest machine representable number (see X02ALF) and ZLG(2) contains zero.

7 Accuracy

If IFAIL = 0 on exit, then the computed roots should be accurate to within a small multiple of the **machine precision** except when underflow (or overflow) occurs, in which case the true roots are within a small multiple of the underflow (or overflow) threshold of the machine.

8 Further Comments

None.

9 Example

To find the roots of the quadratic equation $z^2 + 3z - 10 = 0$.

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      C02AJF Example Program Text
*      Mark 14 Release.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      real              ZERO
      PARAMETER        (ZERO=0.0e0)
*      .. Local Scalars ..
      real              A, B, C
      INTEGER          IFAIL
*      .. Local Arrays ..
      real              ZLG(2), ZSM(2)
*      .. External Subroutines ..
      EXTERNAL         C02AJF
*      .. Intrinsic Functions ..
      INTRINSIC        ABS
*      .. Executable Statements ..
      WRITE (NOUT,*) 'C02AJF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) A, B, C
      IFAIL = 0

*
      CALL C02AJF(A,B,C,ZSM,ZLG,IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Roots of quadratic equation'
      WRITE (NOUT,*)
      IF (ZSM(2).EQ.ZERO) THEN
*          2 real roots.
          WRITE (NOUT,99999) 'z = ', ZSM(1)
          WRITE (NOUT,99999) 'z = ', ZLG(1)
      ELSE
*          2 complex roots.
          WRITE (NOUT,99998) 'z = ', ZSM(1), ' +/- ', ABS(ZSM(2)), '*i'
      END IF
      STOP
*
99999 FORMAT (1X,A,1P,e12.4)
99998 FORMAT (1X,A,1P,e12.4,A,e12.4,A)
      END
```

9.2 Program Data

```
C02AJF Example Program Data
1.0    3.0   -10.0           :A  B  C
```

9.3 Program Results

C02AJF Example Program Results

Roots of quadratic equation

```
z =    2.0000E+00
z =   -5.0000E+00
```